EXTENDED SOLUTIONS TO MOON INFALL PROBLEM

Numerical Solution

Trial and Error Approach:

First, I will present a straightforward computer simulation of the lunar infall problem. Knowing that the time of infall is on the order of days, I will set my step-size to be h=60; we can investigate the reasonableness of this choice later. Writing code that should be very familiar by now:

```
Clear[a, v, x, h, newt, mass]
x[0] = 3.84 × 10^8; mass = 6 × 10^24; newt = 6.67 × 10^(-11);
h = 60;
v[0] = 0;
a[x_] := a[x] = -newt mass / x^2
v[n_] := v[n] = v[n - 1] + a[x[n - 1]] h
x[n_] := x[n] = x[n - 1] + (v[n] + v[n - 1]) h / 2
ListPlot[Table[{n / 1440, x[n]}, {n, 7200}]]
```

Notice that I set the initial velocity to zero, and that I am using "newt" to represent the Newtonian gravitational constant and mass to represent the "mass" of the Earth.   Since we taking a trial and error approach to finding the time of infall (and remembering that our time step is 1 minute), let' s look at a plot for the moon infall problem for a period of five days, or through 7200 iterations :



Ok, we overshot our mark, or our planet, but we are in the ball park. Notice that by plotting n/1440, or the number of minutes/minutes/day, the horizontal axis measures time elapsed in days. So, let' s focus in the area around 4.8 days, or n = 6912. Let' s print out a few values of x to see when x < 0 :

```
Print[x[6920], "  ", x[6950], "  ", x[6980]]
```

2.29353×10^{7} 1.08154×10^{7} -8.06537×10^{7}

We can see that the moon crosses the x = 0 boundary somewhere between n = 6950 and n = 6980; we can continue this process until we find that :

```
Print[x[6964], "  ", x[6965]]
```

1.14504×10^{6} $-698776.$

Or collision occurs between the 6964 th and 6965 th minute, or when t = 4.84 days.

Catch and Throw ... No need for trial and error :

We can determine the smallest value of n for which x[n]<0 by adding one line of code to our program:

```
Clear[a, v, x, h, newt, mass, nterms]
x[0] = 3.84 × 10^8; mass = 6 × 10^24; newt = 6.67 × 10^(-11);
h = 60;
v[0] = 0;
a[x_] := a[x] = -newt mass / x^2
v[n_] := v[n] = v[n - 1] + a[x[n - 1]] h
x[n_] := x[n] = x[n - 1] + (v[n] + v[n - 1]) h / 2
nterms = Catch[Do[If[x[n] < 0, Throw[n - 1]], {n, 7200}]];
Print["The moon crashes into the Earth after the ", nterms, "th iteration."]
```

```
The moon crashes into the Earth after the 6964th iteration.
```

What does the next to last line do? Notice that there is a Do loop set up to iterate 7200 times, even though we know that the moon will have crashed into the earth before then. Additionally, we create an If statement that tests for the condition that x[n] < 0, or that the moon has already crashed. When this statement is true, we execute the "Throw" command. "Throw" terminates the evaluation of the Do loop once x[n] < 0, and stores the most recently computed values. By "Throwing" the (n-1) term, we will be "Catching" the final values for the Moon just prior to impact into the Earth.

The use of "Catch" and "Throw" allow us to consider all kinds of interesting wrinkles to this problem.

Changing the limits of the numerical integration

In class, a student (RA) asked if we should compute when the moon reached x = 0, or if it would be more meaningful to calculate when the moon first hit the surface of the Earth. We can do the latter calculation easily by setting the radius of the earth equal to re and changing the If statement accordingly :

```
Clear[a, v, x, h, newt, mass, re, nterms]
x[0] = 3.84 × 10^8; mass = 6 × 10^24; newt = 6.67 × 10^(-11);
h = 60; re = 6.4 × 10^6;
v[0] = 0;
a[x_] := a[x] = -newt mass / x^2
v[n_] := v[n] = v[n - 1] + a[x[n - 1]] h
x[n_] := x[n] = x[n - 1] + (v[n] + v[n - 1]) h / 2
nterms = Catch[Do[If[x[n] < re, Throw[n - 1]], {n, 7200}]]
```

```
6957
```

And we see that this shaves 7 minutes off the time. But wait, the moon is not a point mass, it also has a radius. If the initial distance provided for x[0] is the distance between the centers of the planets, then we want to calculate when the leading edge of the moon first collides with the surface of the Earth. I will define the radius of the moon as rm and set rm = re/3.7 since the Earth' s radius is 3.7 times that of the moon :

```
Clear[a, v, x, h, newt, mass, re, rm, nterms]
x[0] = 3.84 × 10^8; mass = 6 × 10^24; newt = 6.67 × 10^(-11);
h = 60; re = 6.4 × 10^6; rm = re / 3.7;
v[0] = 0;
a[x_] := a[x] = -newt mass / x^2
v[n_] := v[n] = v[n - 1] + a[x[n - 1]] h
x[n_] := x[n] = x[n - 1] + (v[n] + v[n - 1]) h / 2
nterms = Catch[Do[If[x[n] < (re + rm), Throw[n - 1]], {n, 7200}]]
```

```
6954
```

And that hastens our demise by 3 minutes.

Graphs and Results :

Let' s plot the graphs for x (t) and v (t), using n = 6955 for our number of iterations (this calculates the values for the moon at x <(re+rm)).

```
ListPlot[Table[{n / 1440, x[n]}, {n, nterms}],
 PlotLabel → "Distance profile of infalling moon",
 AxesLabel → {"distance (m)", "time (days)"}, PlotRange → All]
```



Similarly for the velocity curve (note that I am plotting the absolute magnitude of velocity) :

```
ListPlot[Table[{n / 1440, Abs[v[n]]}, {n, nterms}],
 PlotRange → All, PlotLabel → "Velocity profile for infalling moon.",
 AxesLabel → {"Time in days", "Velocity (m/s)"}]
```



Notice that I append the PlotRange -> All statement to get all the points at the end. You may not have known to do this and I will not take off credit if you did not get all the velocity points plotted. Notice how the velocity profile looks almost linear through most of the trip, and velocity increasing rapidly in the last few hours before impact.

Motion in first and last intervals:

The method of using "Catch" and "Throw" to find the value of nterms makes it easy for us to find the values of position and velocity in the moments before . To find the distance traveled in the last minute, we simply take the difference in position between the last and next to last evaluation points (remember, we set h = 60 s) :

```
x[nterms - 1] - x[nterms]
```

```
553 406.
```

Since this is the distance traveled in 60 s, the average speed in this interval is :

```
% / 60
```

```
9223.43
```

or 9.2 km/s.

We can easily find the distance traveled in the first minute :

```
x[0] - x[1]
```

```
4.88525
```

Not quite 5 meters; the longest journey of a quarter of a million miles starts with a single iteration ...

How long did it take the moon to fall 1/2 the distance?

This may seem like a problem you have to do by trial and error, and of course, that will work fine and if done properly, will receive full credit. But we can utilize "Catch" and "Throw" again to find the value of n when the moon is half way to the Earth. Consider the If statement in the following command :

```
Catch[Do[If[Abs[x[n] - x[0] / 2] < 10^5,
   Throw[{n, x[n], Abs[x[n] - x[0] / 2]}]], {n, nterms}]]
```

$\{5698, 1.92026\times 10^{8}, 26\,043.7\}$

The statement computes the absolute value of the distance between the moon and the midway point to the Earth, which is simply one half of the initial distance, or x[0]/2. If the distance is less than 100, 000 m, we abort the calculation and return the value of n at which the moon reaches this point. A tolerance of 100, 000 m may sound like a lot, but it is less than 0.001 of the initial Earth - moon distance, and is smaller than the diameter of Mare Imbrium, a large impact basin on the moon.

We see that the moon reaches the midway point in the 5698 th step (of 6954 steps), or when it has completed 81.9 % of the time needed to reach the Earth. So it makes the first half of the journey in 3.96 days, and the second half in 0.87 days.

Analytic Approaches

We can approach this problem in a couple of more analytic ways to check our results.

Method I: We start very simply with the definition of velocity :

$$v = \frac{dr}{dt} \Rightarrow dt = \frac{dr}{v} \Rightarrow t = \int_R^0 \frac{dr}{v} \tag{1}$$

where R is the distance from the Earth to the moon. So if we can find an expression for v (r), we can substitute it into the definite integral above and find the time it takes for the moon to fall into the Earth.

But how do we find that expression for v(r)? We start with the definition of acceleration, a=dv/dt, along with the fact that the acceleration is caused by the Earth's gravity:

$$a = \frac{dv}{dt} = \frac{-GM}{r^2} \tag{2}$$

where G is the Newtonian Gravitational constant, M is the mass of the Earth, and r is the instantaneous distance of the infalling moon from the Earth. To convert this into an expression for v (r), we use the chain rule plus the definition of velocity (v = dr/dt) to get :

$$a = \frac{dv}{dt} = \frac{dv}{dr}\frac{dr}{dt} = v\frac{dv}{dr} \tag{3}$$

Equating the two expressions for acceleration in eqs. (2) and (3) gives us a very easy differential equation to solve :

$$v\frac{dv}{dr} = \frac{-GM}{r^2} \tag{4}$$

Separating variables and integrating :

$$v^2 = 2\frac{GM}{r} + C \tag{5}$$

where C is a constant of integration. We evaluate C by using the initial condition that v = 0 when r = R:

$$0 = 2\frac{GM}{R} + C \Rightarrow C = -2\frac{GM}{R} \tag{6}$$

and we can write our velocity expression as :

$$v(r) = \sqrt{2\,GM\left(\frac{1}{r} - \frac{1}{R}\right)} \tag{7}$$

Substituting this into the integral for time in equation (1) :

$$t = \int_R^0 \frac{dr}{\sqrt{2\,GM}\,\sqrt{\frac{1}{r} - \frac{1}{R}}} = \frac{1}{\sqrt{2\,GM}} \int_R^0 \frac{dr}{\sqrt{\frac{1}{r} - \frac{1}{R}}} \tag{8}$$

All we have to do is determine the integral in eq. (8). This is a pretty ugly integral, but becomes remarkably tractable with the substitution :

$$r = R \sin^2 \theta \tag{9}$$

For this substitution, we have that

$$dr \;=\; 2\,R \sin\theta \cos\theta \tag{10}$$

Making these substitutions and doing a little bit of algebra allows us to transform the integral in eq. (8) to :

$$t = \frac{2\,R^{3/2}}{\sqrt{2\,G\,M}} \int_{-\pi/2}^{0} \sin^2\theta \, d\theta \;=\; \frac{\pi\,R^{3/2}}{2\,\sqrt{2\,G\,M}} \tag{11}$$

Evaluating this with the known values of the constants :

```
rm = 3.84 × 10^8; newt = 6.67 × 10^(-11); mass = 6 × 10^24;
π rm^(3 / 2) / (2 Sqrt[2 newt mass]) / 86 400
4.83559
```

Dividing by 86, 400 secs/day yields the result of 4.84 days for the moon to fall into the Earth; a result in excellent agreement with our model calculation. The close agreement between this result and the results of our numerical calculations suggests that the use of Euler's method with a step size of 1 minute is sufficient to reproduce the theoretical results.

Method II : Starting with Kepler' s Third Law

If you studied planetary motion and Kepler' s Laws in introductory physics, you might recognize equation (11) as a form of Kepler' s third law which states that the period of orbit around a central mass M is proportional to the 3/2 power of the semi - major axis of the orbit. The equation you may have studied is :

$$MP^2 \;=\; \frac{4\,\pi^2}{G}\, a^3 \tag{12}$$

where M is the mass of the central object (in our case, the Earth), P is the period, G the Newtonian gravitational constant and a the semi - major axis. Writing this in terms of P, we have :

$$P = \frac{2\pi a^{3/2}}{\sqrt{G M}} \tag{13}$$

This describes the time it takes an object to make a complete orbit around its primary. Now, in the case we are considering, we can still consider that the moon is in orbit around the Earth. Instead of a circular (or nearly circular) orbit, this orbit is highly elliptical, such that the Earth is at one end of the orbit. If *a* is the semi - major axis of the "normal" orbit, then *a/2* is the semi - major axis of the infalling orbit. If we substitute *a/2* for *a* in equation (13) we get :

$$P_{\text{infall}} = \frac{2\pi (a/2)^{3/2}}{\sqrt{G M}} = \frac{2\pi a^{3/2}}{\sqrt{8 G M}} = \frac{\pi a^{3/2}}{\sqrt{2 G M}} \tag{14}$$

However, the period we just computed is the time for one complete orbit starting from the current position of the moon, swinging around the Earth and returning to orbital radius of the moon. Of course, if the moon crashes into the earth, it never makes the return trip, so the time it takes to crash is 1/2 of the time in eq. (14) or

$$P_{\text{impact}} = \frac{1}{2} P_{\text{infall}} = \frac{\pi a^{3/2}}{2\sqrt{2 G M}} \tag{15}$$

Compare eq. (15) with eq. (13) and observe that

$$\frac{P_{\text{impact}}}{P} = \frac{1}{4\sqrt{2}} \Rightarrow P_{\text{impact}} = \frac{P}{4\sqrt{2}} \tag{16}$$

The "normal" period of the moon around the Earth, the sidereal period, is 27.32 days, so that this analysis gives us that the time to impact is 4.83 days, consistent with our previous results.